

---

Faculty of Applied Mathematics

---

University of Twente

University for Technical and Social Sciences

---

---

P.O. Box 217  
7500 AE Enschede  
The Netherlands  
Phone +31-53-893400  
Fax +31-53-356695  
Telex 44200

---

MEMORANDUM No. 1100

The INTEGRATOR package  
for REDUCE. Version 1.0

G.H.M. ROELOFS

NOVEMBER 1992

ISSN 0169-2690

# THE INTEGRATOR PACKAGE FOR REDUCE

Version 1.0

MARCEL ROELOFS

**Abstract:** We give the **WEB** source of the **INTEGRATOR** package for **REDUCE**. The package can be used to solve overdetermined systems of partial differential equations, especially those occurring in the computation of symmetries and prolongation structures of (supersymmetric) systems of partial differential equations. The package is based on a former package by Kersten.

**AMS subject classification (1991):** 35N99, 58G37, 68N99, 68Q40, 70H33.

**Keywords:** overdetermined systems, computer algebra software, symmetries, prolongation structures.

	Section	Page
<b>Introduction</b> .....	1	1
<b>Integration of overdetermined systems of p.d.e.'s</b> .....	6	2
Initializing an equation set .....	9	4
The integration procedure .....	15	6
Homogeneous integration .....	17	7
Splitting polynomial equations .....	29	11
Solving Lie algebra relations .....	37	13
Solving a function .....	39	14
Inhomogeneous integration .....	43	16
Generation of new equations by differentiation .....	53	20
<b>Additional tools</b> .....	60	23
<b>Index</b> .....	70	27

**1. Introduction.** In this WEB file we shall describe a REDUCE package for the integration of over-determined systems of partial differential equations (p.d.e.'s). This work is mainly based on a similar package by Paul Kersten for just the determination of symmetry groups (CWI tract 34, *Infinitesimal symmetries: a computational approach*, CWI, Amsterdam, 1987) and an extension by myself which also allows the determination of Wahlquist and Estabrook prolongation algebras.

The main reasons for the implementation of this package, are our improved insight in the internals of REDUCE, the wish to have one combined integrator for both cases and the availability of substantially improved versions of some the procedures used in the former packages.

The “banner line” defined here is intended for identification purposes on loading. It should be changed whenever this file is modified. System dependent changes, however, should be made in a separate change file.

```
define banner ≡ "Integrator_package_for_REDUCE_3.4, $Revision: 1.0 $"
```

**2.** We define the following macros for clarity.

```
define change_to_symbolic_mode ≡ symbolic
define change_to_algebraic_mode ≡ algebraic
define stop_with_error(string_1, expr_1, string_2, expr_2) ≡
    msgpri(string_1, expr_1, string_2, expr_2, t)
define message(string_1, expr_1, string_2, expr_2) ≡
    msgpri(string_1, expr_1, string_2, expr_2, nil)
define operator_name_of ≡ car
define arguments_of ≡ cdr
define first_argument_of ≡ cadr
define second_argument_of ≡ caddr
define first_element_of ≡ car
define rest_of ≡ cdr
define skip_list ≡ cdr { Skip the 'list in front of an algebraic list }
format function = identifier
```

**3.** The following macros are intended as common programming idioms.

```
define incr(x) ≡ (x := x + 1)
define decr(x) ≡ (x := x - 1)
```

**4.** A new REDUCE switch can be introduced using the following code.

```
define initialize_global(global_name, value) ≡
    global '(global_name)$
    global_name := value
define initialize_fluid(fluid_name, value) ≡
    fluid '(fluid_name)$
    fluid_name := value
define new_switch(switch_name, value) ≡
    initialize_fluid(!* Q&switch_name, value)$
    flag('(switch_name), 'switch)
```

**5.** We do all initializations in the beginning of the package.

```
change_to_symbolic_mode$
write banner$ terpri()$
⟨Lisp initializations 9⟩
change_to_algebraic_mode$
```

**6. Integration of overdetermined systems of p.d.e.'s.** For the determination of symmetry groups or prolongation structures of (systems of) partial differential equations, the defining relations give rise to an overdetermined system of p.d.e.'s. Finding the symmetry group or prolongation structure boils down to solving such a system.

There are, however, some differences between the determination of a symmetry group or the determination of a prolongation structure. These differences are:

1. The differential equations for the determination of the symmetry group are linear, the equations for the determination of a prolongation structure are nonlinear. This nonlinearity, however, is of a special kind, namely, the only occurring nonlinear terms are (possibly nested) liebrackets of the functions to be integrated.
2. For the determination of symmetry groups, the functions to be determined integrate to polynomials with constant coefficients. For the determination of prolongation structures, functions integrate to polynomials, coefficients of which are generators of some unknown Lie algebra. The defining relations of this algebra are the remaining (nonlinear) relations which have no dependency on the independent variables involved.

From the above it is clear that integration has to be treated slightly different in either of the cases. The differences are however small enough to allow the implementation of one integrator for both cases.

7. In order to explain all possible p.d.e.'s which can be integrated, we make the following assumptions:

1. Functions are represented by expressions  $f(n)$ , where  $f$  is some specified operator and  $n$  is an integer. Since we intend to use the package for computations for supersymmetric p.d.e.'s, we shall use the notion the elements with  $n$  positive must integrate to an even polynomial and elements with negative  $n$  must integrate to an odd polynomial (this is only useful for computations in prolongation theory, where coefficients can be even or odd Lie algebra generators).
2. The dependencies of functions are solely listed on the dependency list, i.e. must be stated by the 'depend' statement of REDUCE. Notice, however, that we do not allow dependencies of odd variables. The reason for this is a pragmatic one: due to the anticommutivity of odd variables,  $n$  odd variables can only produce  $2^n$  different terms containing these variables, hence can be stated explicitly provided that  $n$  is not too big. On the other hand, if we allow dependencies of odd variables, a lot of additional operators have to be implemented to take care of e.g. partial differentiation w.r.t. odd variables.

8. If  $f$  is the operator denoting functions,  $x$  the operator denoting Lie algebra generators (or, in the case of a symmetry group, just constants), then following the description above, a p.d.e. has the following possible terms (any coefficient  $c$  is always some polynomial in the independent variables):

- A. terms of the form  $c_n df(f(n), \dots)$ .
- B. terms of the form  $c_n f(n)$ .
- C. terms of the form  $c_{1,2}[z_1, z_2]$  where  $z_1, z_2$  are either functions  $f(n)$  or Lie algebra generators  $x(n)$ .
- D. terms of the form  $c_n x(n)$ .

These possibilities lead, in a natural way, to the following strategy of solving the p.d.e.'s:

- 1. If there is only one term of type A, we can integrate this equation homogeneously, i.e. give a polynomial expression for  $f(n)$  using the variables involved in the differential term.
- 2. If the p.d.e. is a polynomial in one or more independent variables on which none of the occurring functions depend, all coefficients of this polynomial have to be zero, i.e., the p.d.e. splits up into a set of smaller p.d.e.'s.
- 3. If there are only terms of type C and D we have a Lie algebra relation, which can be solved by the LIESUPER package, if solvable.
- 4. If there is a function of type B depending on all variables occurring in the p.d.e. and not occurring in a term of type A, we can solve for this function.
- 5. If there is one term of type A depending on all variables occurring the p.d.e. and the remaining terms are polynomial in the variables occurring in the derivative, the p.d.e. can be integrated inhomogeneously.
- 6. If there is just one function in the p.d.e. which depends on a variable only occurring polynomially in the rest of the p.d.e., such that the p.d.e. can not be integrated inhomogeneously since the dependencies of the various occurring functions do not match, we can introduce new equations of type 1 by appropriately differentiating the p.d.e.

**9. Initializing an equation set.** The integrator will be implemented in such a way that integration can be performed on different sets of p.d.e.'s at the same time. Different sets of p.d.e.'s will be distinguished by the name of the operator in which they are stored.

For each operator representing a set of p.d.e.'s we must know: the name of the operator(s) representing the functions and the operator that must be used to represent constant coefficients during the integration. If this last operator possesses the indicator *bracketname* we know that the operator is declared as a Lie algebra generator, hence that we are in the prolongation case. In this case the name of the associated liebracket is equal to the value of *bracketname*.

Moreover, we have to know the total number of equations used, in view of the additional equations that may be generated and which must be numbered subsequently. In connection with the integrations taking place we also have to know the number of functions, resp. constants (generators) being in use.

This is all taken care of by the procedure *initialize\_equations*, which assigns to an operator *operator\_name*, the total number of used equations *total\_used*, the list *variable\_list* of all occurring independent variables, the operator *constant\_operator*, elements of which act as constants, and an arbitrary number of operators *function\_operator* acting as functions. *constant\_operator* and each *function\_operator* should be given an algebraic list of the form {operator, number of even elements used, number of odd elements used}.

In order to allow an arbitrary number of parameters we make *initialize\_equations* a *psopfn*. How *psopfn*'s are dealt with internally is explained in the documentation of either the TOOLS package or the LIESUPER package.

```
(Lisp initializations 9) ≡
  put('initialize_equations, 'psopfn, 'initialize_equations1)$
```

See also sections 13, 18, 33, 53, and 66.

This code is used in section 5.

10.

```
lisp procedure initialize_equations1 specification_list;
  begin scalar operator_name, total_used, variable_list, specification, even_used, odd_used,
    constant_operator, bracketname, function_name, function_list;
  if length specification_list < 5 then
    rederr("INITIALIZE_EQUATIONS: wrong number of parameters");
  if ~idp(operator_name := first_element_of specification_list) then
    rederr("INITIALIZE_EQUATIONS: equations operator must be identifier");
  if ~fixp(total_used := reval first_element_of (specification_list := rest_of
    specification_list)) ∨ total_used < 0 then
    rederr("INITIALIZE_EQUATIONS: total number of equations must be positive");
  put(operator_name, 'total_used, total_used);
  variable_list := reval first_element_of (specification_list := rest_of specification_list);
  if atom variable_list ∨ operator_name_of variable_list ≠ 'list then
    rederr("INITIALIZE_EQUATIONS: variable list must be algebraic list");
  put(operator_name, 'variable_list, skip_list variable_list);
  { Check and initialize constant_operator 11 };
  { Check and initialize function_list 12 };
end$
```

11. The *constant\_operator* can either be a Lie algebra generator or not. If so, we also have to assign the associated liebracket to *operator\_name* and used the procedure *define\_used* to take care of the assignment of the used dimensions to the liebracket. If *constant\_operator* is not a Lie algebra generator, we store these dimensions in the same way as happens for liebrackets.

```

define check_valid_function_declaration(op_list, op_name) ≡
  if atom op_list ∨ length op_list ≠ 4 ∨ operator_name_of op_list ≠ 'list
    ∨ ¬idp(op_name := first_argument_of op_list) ∨ ¬fixp(even_used := reval caddr op_list)
    ∨ ¬fixp(odd_used := reval caddr op_list) ∨ even_used < 0 ∨ odd_used < 0 then
    stop_with_error("INITIALIZE_EQUATIONS:␣invalid␣declaration␣of", op_list, nil, nil)
define put_used_dimensions(op_name, even_used, odd_used) ≡
  if get(op_name, 'bracketname) then define_used(bracketname, list('list, even_used, odd_used))
  else
    begin put(op_name, 'even_used, even_used);
      put(op_name, 'odd_used, odd_used);
    end
⟨ Check and initialize constant_operator 11 ⟩ ≡
  specification_list := rest_of specification_list; specification := first_element_of specification_list;
  check_valid_function_declaration(specification, constant_operator);
  put(operator_name, 'constant_operator, constant_operator);
  if (bracketname := get(constant_operator, 'bracketname)) then
    put(operator_name, 'bracketname, bracketname);
  put_used_dimensions(constant_operator, even_used, odd_used)

```

This code is used in section 10.

12.

```

⟨ Check and initialize function_list 12 ⟩ ≡
  for each function_specification in rest_of specification_list do
    begin check_valid_function_declaration(function_specification, function_name);
      put_used_dimensions(function_name, even_used, odd_used);
      function_list := function_name . function_list;
    end;
  put(operator_name, 'function_list, function_list)

```

This code is used in section 10.

13. Since we can apparently choose different sets of p.d.e.'s for solving, we must tell the integrator which set to take. This is done via a global variable *cur\_eq\_set!\**. We will take the operator *equ* as the default *cur\_eq\_set!\**. In this file we will use the abbreviation *ces!\** for *cur\_eq\_set!\**.

```

define ces!* ≡ cur_eq_set!*
⟨ Lisp initializations 9 ⟩ + ≡
  initialize_global(ces!*, 'equ)$

```

14.

```

lisp operator use_equations;
lisp procedure use_equations operator_name;
  begin
    if idp operator_name then ces!* := operator_name
    else rederr("USE_EQUATIONS:␣argument␣must␣be␣identifier");
  end$

```

**15. The integration procedure.** The implementation of the integrator follows the description of all the possible steps given above.

For the use of the fluid variable *listpri\_depth!\**, see below. Its local rebinding is necessary for a proper printing of the messages given by the procedure.

```

lisp operator integrate_equation;
lisp procedure integrate_equation n;
  begin scalar listpri_depth!*, total_used, equation, denominator, solvable_kernel, solvable_kernels,
    df_list, function_list, present_functions_list, variable_list, absent_variables, linear_functions_list,
    constants_list, bracketname, df_terms, df_functions, linear_functions, functions_and_constants_list,
    commutator_functions, present_variables, nr_of_variables, integration_variables;
  listpri_depth!* := 200; terpri!* t;
  (Find the equation to be integrated 16);
  (Step 1: search for homogeneous integration 20);
  (Step 2: search for polynomial behaviour 29);
  (Step 3: search for a Lie relation 37);
  (Step 4: search for a solvable function 39);
  (Step 5: search for inhomogeneous integration 43);
  (Step 6: search for a useful differentiation 54);
  (Step 7: print a "Not solved" message 59);
  solved: { Go here when the equation is solved or its type is determined }
  end$

```

**16.** The part of the equation containing all necessary information is its numerator. For reasons that will become clear in the sequel we need, however, also know its denominator. If the equation is zero, no analysis has to be performed.

```

define nullify_equation(n) ≡
  setk(list(ces!*, n), 0)
(Find the equation to be integrated 16) ≡
  if null(total_used := get(ces!*, 'total_used)) ∨ n > total_used then
    stop_with_error("INTEGRATE_EQUATIONS: properly initialize", ces!*, nil, nil);
  if null(equation := cadr assoc(list(ces!*, n), get(ces!*, 'kvalue))) then
    stop_with_error("INTEGRATE_EQUATION:", list(ces!*, n), "is non-existent", nil);
  denominator := denr(equation := simp!* equation); equation := numr equation;
  if null equation then
    ≪ write ces!*, "(" n, ")" = 0"; terpri!* t; nullify_equation(n); goto solved ≫

```

This code is used in section 15.



**17. Homogeneous integration.** Homogeneous integration must be performed if the equation consists of just one *df* term. In order to find all possible *df* terms we apply *split\_form* to *equation*. This returns a list the *car* of which is the part of *equation* independent of the *df* operator, the *cdr* of which is a list of all linear *df* terms, together with their coefficients. *split\_form* will return with an error if nonlinear *df* terms occur.

```
define independent_part_of ≡ car
define kc_list_of ≡ cdr
define kernel_of ≡ car    { For use with a kernel-coefficient list }
define coefficient_of ≡ cdr { For use with a kernel-coefficient list }
```

**18.** If there is one *df* term, we only solve it if its coefficient is a number, by default. This behaviour is governed by the switch *coefficient\_check*, which is **on** by default. In order to check the coefficient we will use the procedure *find\_solvable\_kernel* to be explained below.

```
(Lisp initializations 9) + ≡
  new_switch(coefficient_check, t)$
```

**19.** Before continuing we introduce some auxiliary macros and procedures.

```
define assoc_delete(kernel, assoc_list) ≡
  delete(assoc(kernel, assoc_list), assoc_list)

lisp procedure successful_message_for(n, action, kernel);
  << write ces!*, "(", n, "):␣", action; maprin kernel; terpri!*(¬!*nat);
  nullify_equation(n); t >>$

lisp procedure not_a_number_message_for(n, action, kernel);
  << write "***␣", ces!*, "(", n, "):␣", action, "␣failed:"; terpri!* t;
  write "␣␣␣␣coefficient␣not␣a␣number␣for␣"; maprin kernel; terpri!*(¬!*nat);
  write "␣␣␣␣Solvable␣with␣'off␣coefficient_check'";
  terpri!* t; t >>$
```

**20.**

```
(Step 1: search for homogeneous integration 20) ≡
  df_list := split_form(equation, '(df));
  if try_a_homogeneous_integration(n, denominator, df_list) then goto solved
```

This code is used in section 15.

**21.**

```
lisp procedure try_a_homogeneous_integration(n, denominator, df_list);
  begin scalar solvable_kernel, solvable_kernels, df_kernel;
  return
    if null independent_part_of df_list ∧ (kc_list_of df_list) ∧ length(kc_list_of df_list) = 1 then
      if (solvable_kernel := find_solvable_kernel(
        solvable_kernels := list(kernel_of first_element_of kc_list_of df_list),
        kc_list_of df_list, denominator)) then
        << df_kernel := first_argument_of solvable_kernel;
        setk(df_kernel, homogeneous_integration_of(solvable_kernel));
        depl!* := assoc_delete(df_kernel, depl!*); { Remove df_kernel from the depl!* list }
        successful_message_for(n, "Homogeneous␣integration␣of␣", solvable_kernel) >>
      else not_a_number_message_for(n, "Homogeneous␣integration", first_element_of solvable_kernels)
  end$
```

22. The procedure *find\_solvable\_kernel* tries to find the first element of *kernel\_list* which has a number as coefficient. If *coefficient\_check* is *off* we can simply take the first element of *kernel\_list*, otherwise we can most conveniently implement a recursive procedure *first\_solvable\_kernel*, which finds the first element of *kernel\_list* with a number as coefficient. We can check this by first checking if the numerator of the coefficient is a domain element, or if the whole coefficient is a number.

```

lisp procedure find_solvable_kernel(kernel_list, kc_list, denominator);
  if !*coefficient_check then first_solvable_kernel(kernel_list, kc_list, denominator)
  else first_element_of kernel_list$

lisp procedure first_solvable_kernel(kernel_list, kc_list, denominator);
  if kernel_list then
    (if domainp coefficient_of kc_pair ∨ numberp !*ff2a(coefficient_of kc_pair, denominator) then
      kernel_of kc_pair
    else first_solvable_kernel(rest_of kernel_list, kc_list, denominator))
    where kc_pair = assoc(first_element_of kernel_list, kc_list)$

```

23. The equation

$$\frac{\partial^{k_1}}{\partial x_1^{k_1}} \cdots \frac{\partial^{k_m}}{\partial x_m^{k_m}} f(x_1, \dots, x_n) = 0 \quad (m \leq n)$$

has general solution

$$f = \sum_{j=1}^m \sum_{i_j=0}^{k_j-1} x_j^{i_j} f_{j,i_j}(x_1, \dots, \hat{x}_j, \dots, x_n).$$

Thus, given a homogenous p.d.e., *homogeneous\_integration\_of* has to return the REDUCE equivalent of the last expression.

If *f* depends on only one variable the  $f_{j,i_j}$  are constants, otherwise they are new functions with dependency on one less variable. In the Lie algebra case the constants are generators of the Lie algebra. Since the dimensions of a *liebracket* in REDUCE have to be given on beforehand, there may not be enough generators left to generate *f*. In this case, we have to enlarge the *liebracket*.

```

define get_dependencies_of(kernel) ≡
  ((if depl_entry then cdr depl_entry)
   where depl_entry = assoc(kernel, depl!*))

lisp procedure homogeneous_integration_of df_term;
  begin scalar df_function, function_number, dependency_list, integration_list, coefficient_name,
    bracketname, even_used, odd_used, integration_variable,
    number_of_integrations, solution, new_dependency_list;
  { Check if df_term can be integrated, find df_function and function_number 24 };
  dependency_list := get_dependencies_of(df_function);
  if length dependency_list = 1 then coefficient_name := get(ces!*, 'constant_operator)
  else coefficient_name := operator_name_of df_function;
  { Get even_used, odd_used and if necessary bracketname 25 };
  integration_list := rest_of arguments_of df_term;
  { Find the next integration_variable and number_of_integrations 26 };
  if bracketname then { Check and possibly enlarge dimensions of bracketname 27 };
  { Perform the integration 28 };
  return solution
end$

```

24. We required *df\_term* to be of the form  $df(f(k), \dots)$  where  $f$  is a function occurring on the *function\_list* of *ces!\** and  $k$  is an integer not equal to zero.

```

{ Check if df_term can be integrated, find df_function and function_number 24 } ≡
  df_function := first_argument_of df_term;
  if ¬member(operator_name_of df_function, get(ces!*, 'function_list'))
    ∨ ¬fixp(function_number := first_argument_of df_function) ∨ function_number = 0 then
    stop_with_error("PERFORM_HOMOGENEOUS_INTEGRATION:␣integration␣of", df_function,
      "not␣allowed", nil)

```

This code is used in section 23.

25. In the liebracket case *even\_used* and *odd\_used* are stored as properties of *bracketname* instead of *coefficient\_name*.

```

{ Get even_used, odd_used and if necessary bracketname 25 } ≡
  if (bracketname := get(coefficient_name, 'bracketname')) then
    begin even_used := get(bracketname, 'even_used');
          odd_used := get(bracketname, 'odd_used');
        end
  else
    begin even_used := get(coefficient_name, 'even_used');
          odd_used := get(coefficient_name, 'odd_used');
        end

```

This code is used in section 23.

26. Finding the integration variables is rather straightforward.

```

{ Find the next integration_variable and number_of_integrations 26 } ≡
  if integration_list then integration_variable := first_element_of integration_list
  else integration_variable := nil;
  if integration_variable ∧ (integration_list := rest_of integration_list)
    ∧ fixp first_element_of integration_list then
    << number_of_integrations := first_element_of integration_list;
      integration_list := rest_of integration_list >>
  else number_of_integrations := 1

```

This code is used in sections 23 and 28.

27. If *df\_function* depends on only one variable, the number of constants being introduced is equal to the *number\_of\_integrations*. The even and odd dimension of *bracketname* are stored as the properties *even\_dimension* and *odd\_dimension*.

```

{ Check and possibly enlarge dimensions of bracketname 27 } ≡
  if function_number > 0 then
    (if even_used + number_of_integrations > get(bracketname, 'even_dimension') then
      change_dimensions_of(bracketname, even_used + number_of_integrations,
        get(bracketname, 'odd_dimension'))
    else
      (if odd_used + number_of_integrations > get(bracketname, 'odd_dimension') then
        change_dimensions_of(bracketname, get(bracketname, 'even_dimension'),
          odd_used + number_of_integrations))

```

This code is used in section 23.

**28.** The actual integration is fairly straightforward by now: for all the possible integration variables we can simply add new terms to *solution*.

```

define new_coefficient ≡
    list(coefficient_name,
        if function_number > 0 then incr(even_used)
        else -incr(odd_used))
define ext_mksq(kernel, power) ≡
    if power = 0 then 1 ./ 1
    else mksq(kernel, power)
define depend_new_coefficient(dependency_list) ≡
    depl!* := (list(coefficient_name,
        if function_number > 0 then even_used
        else -odd_used) . dependency_list) . depl!*;
⟨ Perform the integration 28 ⟩ ≡
    solution := nil ./ 1;
    while integration_variable do
        begin new_dependency_list := delete(integration_variable, dependency_list);
        for i := 0: number_of_integrations - 1 do
            << solution := addsq(solution, multsq(ext_mksq(integration_variable, i), mksq(new_coefficient, 1)));
            if new_dependency_list then depend_new_coefficient(new_dependency_list) >>;
        ⟨ Find the next integration_variable and number_of_integrations 26 ⟩
        end;
    solution := mk!*sq subs2 solution;
    put_used_dimensions(coefficients_name, even_used, odd_used)

```

This code is used in section 23.

**29. Splitting polynomial equations.** For the polynomial behaviour of *equation* we need to know the dependencies of all the functions occurring in *equation* at any level. If there occur any other variables in *equation* and *equation* is polynomial in these variables, the coefficients of this polynomial give rise to a new set of equations.

```
define pc_list_of ≡ kc_list_of { power-coefficient list }
define powers_of ≡ kernel_of
⟨ Step 2: search for polynomial behaviour 29 ⟩ ≡
  ⟨ Find present_functions_list and the absent_variables 30 ⟩;
  if split_equation_polynomially(n, total_used, equation, absent_variables) then goto solved
```

This code is used in section 15.

**30.** Finding all the functions in *equation* can be done by applying the procedure *get\_recursive\_kernels* of the TOOLS package.

```
⟨ Find present_functions_list and the absent_variables 30 ⟩ ≡
  function_list := get(ces!*, 'function_list');
  present_functions_list := get_recursive_kernels(equation, function_list);
  variable_list := get(ces!*, 'variable_list'); absent_variables := variable_list;
  for each function in present_functions_list do
    for each variable in get_dependencies_of(function) do
      absent_variables := delete(variable, absent_variables)
```

This code is used in section 29.

**31.**

```
lisp procedure split_equation_polynomially(n, total_used, equation, absent_variables);
  begin scalar polynomial_variables, equations_list;
  ⟨ Find the polynomial_variables and test for polynomial behaviour 32 ⟩;
  ⟨ If possible, split up equation into smaller equations 35 ⟩
  end$
```

**32.** In most cases the equations under consideration are polynomial in any of the variables and therefore we shall by default not test for polynomial behaviour. This testing is governed by the switch *polynomial\_check* which, by default, is *off*. If it is *on* testing is done by the procedure *polynomialp* to be defined below.

```
⟨ Find the polynomial_variables and test for polynomial behaviour 32 ⟩ ≡
  polynomial_variables := absent_variables;
  if !*polynomial_check then
    polynomial_variables := for each variable in polynomial_variables join
      if polynomialp(equation, variable) then list(variable)
```

This code is used in section 31.

**33.** ⟨ Lisp initializations 9 ⟩ + ≡  
*new\_switch*(*polynomial\_check*, *nil*)\$

**34.** Checking a standard form for polynomial behaviour in some kernel can be done by checking the main variable, the leading coefficient and the reductum, respectively.

```
lisp procedure polynomialp(expression, kernel);
  if domainp expression then t
  else ((main_variable = kernel ∨ ¬depends(main_variable, kernel))
        ∧ polynomialp(lc expression, kernel) ∧ polynomialp(red expression, kernel))
  where main_variable = mvar expression$
```

35. The coefficients of a polynomial can be found by applying the procedure *multi\_split\_form* from the *TOOLS* package.

If *equation* can be split into smaller equations, *split\_equation\_polynomially* has to return *t*.

```
(If possible, split up equation into smaller equations 35) ≡
  equations_list := multi_split_form(equation, polynomial_variables);
  if length equations_list > 1 then
    << for each pc_pair in pc_list_of equations_list do
      setk(list(ces!*, incr(total_used)), mk!*sq((coefficient_of pc_pair) ./ 1));
      if independent_part_of equations_list then
        setk(list(ces!*, incr(total_used)), mk!*sq((independent_part_of equations_list) ./ 1));
        write ces!*, "(" , n, ")" ⊔ breaks ⊔ into ⊔ , ces!*, "(" , get(ces!*, 'total_used') + 1,
          " ) , . . . , " , ces!*, "(" , total_used , ")" ⊔ by ⊔ ;
        maprin partial_list(polynomial_variables, 5); terpri!* (¬!*nat);
        nullify_equation(n); put(ces!*, 'total_used', total_used) >>;
    if length equations_list > 1 then
      return t
```

This code is used in section 31.

36. In order to get messages in a readable form, we sometimes need to print lists partially. This is taken care of the following procedures.

```
lisp procedure partial_list(printed_list, nr_of_items);
  'list . broken_list(printed_list, nr_of_items)$

lisp procedure broken_list(list, n);
  if list then
    if n = 0 then '(!!!)
    else car list . broken_list(cdr list, n - 1)$
```

**37. Solving Lie algebra relations.** If the first two steps have failed, we need to analyze *equation* in a more drastic way: we need to find all functions occurring linearly in *equation*, and if a *liebracket* is specified, all commutators and algebra generators occurring in *equation* as well. Since we have already looked for *df* terms in *equation* in each next step we only have to examine the independent part of the previous step.

```

⟨Step 3: search for a Lie relation 37⟩ ≡
  linear_functions_list := split_form(independent_part_of df_list, function_list);
  df_list := kc_list_of df_list;
  constants_list := split_form(independent_part_of linear_functions_list, list get(ces!*, 'constant_operator'));
  linear_functions_list := kc_list_of linear_functions_list;
  if (bracketname := get(ces!*, 'bracketname')) then ⟨Solve equation if it is a Lie expression 38⟩

```

This code is used in section 15.

**38.** In the Lie algebra case we can try to solve the Lie expression if there are no *df* terms or linearly occurring functions. Solving Lie expression can be done using the procedure *relation\_analysis* of the LIESUPER package. *relation\_analysis* returns either the kernel for which the relation is solved or an atom indicating the nature of the non-solvability.

```

⟨Solve equation if it is a Lie expression 38⟩ ≡
  if length(df_list) = 0 ∧ length(linear_functions_list) = 0 then
    << if atom(solvable_kernel := relation_analysis(!*ff2a(equation, denominator), bracketname)) then
      << write ces!*, "(", n, ")_is_a_non-solvable_Lie_relation"; terpri!* t >>
    else
      << write ces!*, "(", n, ")_solved_for_"; maprin solvable_kernel; terpri!* t;
      nullify_equation(n) >>
    goto solved >>

```

This code is used in section 37.

**39. Solving a function.** If *equation* is not a Lie expression, there may be a function or a constant for which we can solve it. In order to do this we need to

- find all variables *present\_variables*, on which at least one of the present functions *recursive\_functions\_list* depends; of course it is the complement of *absent\_variables* in *variable\_list*.
- find all linearly occurring functions *solvable\_kernels* which depend on all of the *present\_variables*; these are the possible candidates for solving. If there are no *present\_variables*, *equation* is apparently a relation between some constants and we can try to solve one.
- remove all functions from *solvable\_kernels*, which also occur in a *df* term, or in the liebracket case, in a commutator.
- if *coefficient\_check* is **on** we must only solve for those functions which have a number as coefficient. This is checked by the procedure *find\_solvable\_kernel*.

Before doing anything we shall, however, construct lists containing all functions occurring in *df* terms, occurring linearly (and the constants) and, if necessary, occurring in commutators. These lists will also come in handy in the next steps.

```

⟨Step 4: search for a solvable function 39⟩ ≡
  ⟨Construct df_terms, df_functions, linear_functions and commutator_functions 40⟩;
  ⟨Get present_variables and nr_of_variables 41⟩;
  for each kernel in linear_functions do
    if length get_dependencies_of(kernel) = nr_of_variables then
      solvable_kernels := kernel . solvable_kernels;
  for each kernel in append(df_functions, commutator_functions) do
    solvable_kernels := delete(kernel, solvable_kernels);
  if solvable_kernels then ⟨Try to solve a function 42⟩

```

This code is used in section 15.

**40.** Of course we are only interested in *df* terms of functions occurring on *function\_list*.

```

⟨Construct df_terms, df_functions, linear_functions and commutator_functions 40⟩ ≡
  df_terms := for each df_term in df_list join
    if member(operator_name_of first_argument_of kernel_of df_term, function_list) then
      list kernel_of df_term;
  for each df_term in df_terms do
    if ¬member(first_argument_of df_term, df_functions) then
      df_functions := first_argument_of(df_term) . df_functions;
  functions_and_constants_list := append(linear_functions_list, kc_list_of constants_list);
  linear_functions := for each linear_function in functions_and_constants_list collect
    kernel_of linear_function;
  if bracketname then commutator_functions :=
    get_recursive_kernels(independent_part_of constants_list, get(ces!*, 'function_list'));

```

This code is used in section 39.

```

41.  ⟨Get present_variables and nr_of_variables 41⟩ ≡
  present_variables := variable_list;
  for each variable in absent_variables do present_variables := delete(variable, present_variables);
  nr_of_variables := length present_variables

```

This code is used in section 39.



```

42.    ⟨Try to solve a function 42⟩ ≡
    << solvable_kernel := find_solvable_kernel(solvable_kernels, functions_and_constants_list, denominator);
    if solvable_kernel then
        << linear_solve_and_assign(!*ff2a(equation, 1), solvable_kernel);
        depl!* := assoc_delete(solvable_kernel, depl!*);
        { Remove the dependencies of the solved function }
        successful_message_for(n, "Solved_for", solvable_kernel); goto solved >>
    else
        << not_a_number_message_for(n, "Solving_a_function", partial_list(solvable_kernels, 3));
        goto solved >>>

```

This code is used in section 39.

**43. Inhomogeneous integration.** For an inhomogeneous integration, we are looking for a maximal *df* term, i.e. which has dependency on all the *present\_variables*, such that the remaining part of *equation* is polynomial in the variables, w.r.t. which the function in the *df* term is differentiated, i.e. a) we only have to look at *df* terms which are differentiated w.r.t. variables on which none of the non-maximally occurring functions in *equation* depend, and b) if *polynomial\_check* is **on**, we must check explicitly if the rest of *equation* is polynomial in these variables.

We shall collect the list of “integrable” variables in the list *integration\_variables*.

```
{ Step 5: search for inhomogeneous integration 43 } ≡
{ Find the possible integration_variables 44 };
if try_an_inhomogeneous_integration(n, equation, denominator, df_list, df_terms, integration_variables,
  nr_of_variables) then goto solved
```

This code is used in section 15.

**44.** Finding the *integration\_variables* is rather easy using the lists *df\_functions*, *linear\_functions* and *commutator\_functions*. Starting with *present\_variables* we have to delete all variables on which one of the *linear\_functions* or *commutator\_functions* depend, or one of the *df\_functions*, which do not have maximal dependency, i.e. which do not depend on *nr\_of\_variables* variables.

```
{ Find the possible integration_variables 44 } ≡
integration_variables := present_variables;
for each kernel in append(linear_functions, commutator_functions) do
  for each variable in get_dependencies_of(kernel) do
    integration_variables := delete(variable, integration_variables);
for each df_function in df_functions do
  if ¬length get_dependencies_of(df_function) = nr_of_variables then
    for each variable in get_dependencies_of(df_function) do
      integration_variables := delete(variable, integration_variables)
```

This code is used in section 43.

**45.** Finding the integrable *df* terms is rather easy know: find all the *df* terms which have maximal dependency and are only differentiated w.r.t. variables occurring on *integration\_variables*. In order to check this last item we need to know the form of *df* term: it is a list '(*df* function differentiation\_sequence), where *differentiation\_sequence* is a sequence of variables, each variable optionally followed by a integer indicating the number of differentiations w.r.t. to that variable. The procedure *check\_differentiation\_sequence* checks whether all variables in a *differentiation\_sequence* are member of the second argument *variable\_list*.

```
lisp procedure check_differentiation_sequence(sequence, variable_list);
  if null sequence then t
  else if fixp first_element_of sequence ∨ member(first_element_of sequence, variable_list) then
    check_differentiation_sequence(rest_of sequence, variable_list)$
```

**46.**

```
lisp procedure try_an_inhomogeneous_integration(n, equation, denominator, df_list, df_terms,
  integration_variables, nr_of_variables);
  begin scalar solvable_kernel, solvable_kernels, forbidden_functions,
    df_kernel, inhomogeneous_term;
  { Find the integrable df_terms 47 };
  { Find a solvable_kernel, check the inhomogeneous_term and possibly integrate 48 }
  end$
```

47. There one situation we have to take care of specifically: if there are more *df\_terms* for the same function, only one of which is differentiated just w.r.t. *integration\_variables*, we are not allowed to integrate, since the function would be expressed in itself. In this case, we will make *solvable\_kernels* a list of at least length 2 in order to prevent integration.

```

⟨Find the integrable df_terms 47⟩ ≡
  for each df_term in df_terms do
    ≪ if length get_dependencies_of(first_argument_of df_term) = nr_of_variables
      ∧ (check_differentiation_sequence(rest_of arguments_of df_term, integration_variables)
        ∨ member(first_argument_of df_term, forbidden_functions)) then
        solvable_kernels := if member(first_argument_of df_term, forbidden_functions) then list(nil, nil)
        else df_term . solvable_kernels;
        forbidden_functions := (first_argument_of df_term) . forbidden_functions ≫;

```

This code is used in section 46.

```

48.  ⟨Find a solvable_kernel, check the inhomogeneous_term and possibly integrate 48⟩ ≡
  return
  if solvable_kernels then
    if length(solvable_kernels) = 1 then
      if (solvable_kernel := find_solvable_kernel(solvable_kernels, df_list, denominator)) then
        if (inhomogeneous_term := linear_solve(mk!*sq(equation ./ 1), solvable_kernel))
          ∧ (¬!*polynomial_check
            ∨ check_polynomial_integration(solvable_kernel, inhomogeneous_term)) then
          ≪ df_kernel := first_argument_of solvable_kernel;
            setk(df_kernel, inhomogeneous_integration_of(solvable_kernel, inhomogeneous_term));
            depl!* := assoc_delete(df_kernel, depl!*); { Remove df_kernel from the depl!* list }
            successful_message_for(n, "Inhomogeneous␣integration␣of␣", solvable_kernel) ≫
        else
          ≪ write ces!*, "(", n, "):␣Inhomogeneous␣integration␣failed:␣"; terpri!* t;
            write "inhomogeneous␣term␣not␣polynomial␣in␣integration␣variables"; terpri!* t;
            t ≫
        else not_a_number_message_for(n, "Inhomogeneous␣integration",
          first_element_of solvable_kernels)
    else
      ≪ write ces!*, "(", n, "):␣Inhomogeneous␣integration␣failed:␣"; terpri!* t;
        write "more␣terms␣with␣maximal␣dependency"; terpri!* t; t ≫

```

This code is used in section 46.

49. Checking that the inhomogeneous term is polynomial in the integration variables is fairly easy. For all the integration variables we have to check that the denominator does not depend on it and the numerator should be polynomial.

```

lisp procedure check_polynomial_integration(df_term, integration_term);
  begin scalar numerator, denominator, integration_variables, variable, ok;
  numerator := numr simp integration_term; denominator := denr simp integration_term;
  integration_variables := for each argument in rest_of arguments_of df_term join
    if  $\neg$ fixp argument then list argument;
  ok := t;
  while ok  $\wedge$  integration_variables do
     $\ll$  variable := first_element_of integration_variables;
    ok := ( $\neg$ depends(denominator, variable)  $\wedge$  polynomialp(numerator, variable));
    integration_variables := rest_of integration_variables  $\gg$ ;
  return ok;
end$

```

50. We can perform the inhomogeneous integration by applying *multi\_split\_form* to find all the polynomial components of the inhomogeneous term and *homogeneous\_integration\_of* for solving the homogeneous equation.

```

lisp procedure inhomogeneous_integration_of(df_term, inhomogeneous_term);
  begin scalar df_sequence, integration_variables, int_sequence, variable, nr_of_integrations,
    integration_terms, solution, powers, coefficient, int_factor, solution_term, n, k;
  df_sequence := rest_of arguments_of df_term;
  { Find the integration_variables and int_sequence 51 };
  integration_terms := multi_split_form(numr simp inhomogeneous_term, integration_variables);
  integration_terms := (nil . independent_part_of integration_terms) . pc_list_of integration_terms;
  { Make integration_terms a full blown pc_list }
  { Perform the inhomogeneous integration of the numerator of inhomogeneous_term 52 };
  solution := multsq(solution, 1 ./ denr simp inhomogeneous_term);
  solution := mk!*sq subs2 addsq(solution, simp homogeneous_integration_of df_term);
  return solution
end$

```

51. We must analyze *df\_sequence* to get all the integration variables, together with the number of integrations belonging to them.

```

{ Find the integration_variables and int_sequence 51 }  $\equiv$ 
while df_sequence do
   $\ll$  variable := first_element_of df_sequence; df_sequence := rest_of df_sequence;
  if df_sequence  $\wedge$  fixp first_element_of df_sequence then
     $\ll$  nr_of_integrations := first_element_of df_sequence; df_sequence := rest_of df_sequence  $\gg$ 
  else nr_of_integrations := 1;
  integration_variables := variable . integration_variables;
  int_sequence := (variable . nr_of_integrations) . int_sequence  $\gg$ 

```

This code is used in section 50.

52. The particular solution of the equation  $F^{(k)}(x) = x^n$  is

$$F(x) = \frac{1}{(n+1)\cdots(n+k)} x^{n+k}.$$

This process has to be performed for all the terms in *integration\_terms* and for all integrations in *int\_sequence*.

```
( Perform the inhomogeneous integration of the numerator of inhomogeneous_term 52 ) ≡
  solution := nil ./ 1;
  for each term in integration_terms do
    << powers := powers_of term; coefficient := coefficient_of term; int_factor := 1;
    solution_term := 1 ./ 1;
    for each integration in int_sequence do
      << variable := car integration; k := cdr integration;
      n := (if power then cdr power else 0) where power = assoc(variable, powers);
      { If variable does not occur in term, n = 0 }
      for i := 1:k do int_factor := (n + i)*int_factor;
      solution_term := multsq(solution_term, mksq(variable, n + k)) >>;
      solution_term := multsq(solution_term, coefficient ./ int_factor);
      solution := addsq(solution, solution_term) >>
```

This code is used in section 50.

**53. Generation of new equations by differentiation.** As a last method of solving we notice the following: if there is a variable, such that just one *df* term or just one linearly occurring function depends on it and all the other terms are polynomial in this variable, let's say of degree *n*, then we can differentiate *equation* *n* + 1 times to get a new equation of type A.

Experience has proven, however, that applying the above mentioned method, generally will lead to multiple generation of equivalent terms in the answer. Therefore we will only generate a new equation if the switch *allow\_differentiation* is **on**, otherwise we will only generate a message that it is possible to generate a new equation of type A. Solving of such a new equation is always left to the responsibility of the user.

```
(Lisp initializations 9) + ≡
  new_switch(allow_differentiation, nil)$
```

**54.** After this introduction it is clear what we have to do for step 6:

```
(Step 6: search for a useful differentiation 54) ≡
  if try_a_differentiation(n, total_used, equation, present_variables, df_terms, linear_functions,
    commutator_functions) then goto solved
```

This code is used in section 15.

**55.**

```
lisp procedure try_a_differentiation(n, total_used, equation, present_variables, df_terms, linear_functions,
  commutator_functions);
  begin scalar differentiations_list, polynomial_order;
    (Count the number of occurrences of all present_variables 56);
    (If possible and allowed, generate new equations 57)
  end$
```

56. Counting the occurrence of variables is rather easy. For all functions in *df-terms*, *linear-functions* and *commutator-functions*, we have to count the occurrences of all the variables in their respective entries on the dependency list *depl!\**.

For this purpose we rebuild *present-variables* to an association list with entries of the form *variable . origin . number-of-occurrences* where *origin* indicates the *df-term*, *linear-function* or *commutator-function* in which *variable* occurred last.

The action of the following macros, which harmlessly make use of the procedure *rplacd*, is clear.

```

define reinitialize-present-variables ≡
    present-variables := for each variable in present-variables collect (variable . nil . 0)
define variable-of ≡ car
define origin-of ≡ cadr
define counter-of ≡ caddr
define update-variable(variable, origin) ≡
    rplacd(entry, origin . (counter-of entry + 1)) where entry = assoc(variable, present-variables)
define update-variables-using(kernel-list, kernel-selector, flag-function) ≡
    for each kernel in kernel-list do
        for each variable in get-dependencies-of(kernel-selector(kernel)) do
            update-variable(variable, flag-function(kernel));
define identity-function(kernel) ≡ kernel
define empty-function(kernel) ≡ nil
(Count the number of occurrences of all present-variables 56) ≡
    reinitialize-present-variables;
    update-variables-using(df-terms, first-argument-of, identity-function);
    update-variables-using(linear-functions, identity-function, identity-function);
    update-variables-using(commutator-functions, identity-function, empty-function)

```

This code is used in section 55.

57. After the preceding step we can generate new equations by differentiating *equation* w.r.t. to all those variables which occur in only one *df-term* or *linear-function* and for which all other terms of *equation* are polynomial. Using the above code one can check that these variables are exactly the ones for which the *origin* has a value and the *counter* is 1.

```

<If possible and allowed, generate new equations 57> ≡
differentiations_list := for each entry in present_variables join
  if origin_of entry ∧ counter_of entry = 1
    ∧ (polynomial_order :=
      get_polynomial_order(linear_solve(mk!*sq(equation ./ 1), origin_of entry), variable_of entry))
    then list(variable_of entry . origin_of entry . (polynomial_order + 1));
return
if differentiations_list then
  if !*allow_differentiation then
    <<for each entry in differentiations_list do
      setk(list(ces!*, incr(total_used)),
        mk!*sq simpdf list(mk!*sq(equation ./ 1), variable_of entry, counter_of entry));
      write ces!*, "(", n, "):_Generation_of_", ces!*, "(", get(ces!*, 'total_used') + 1, "), ..., ",
        ces!*, "(", total_used, ")_by_differentiation_w.r.t._"; terpri!* t;
      maprin partial.list(for each entry in differentiations_list collect
        list('list, variable_of entry, counter_of entry), 10);
      terpri!*(-!*nat); put(ces!*, 'total_used, total_used); t >>
    else
      <<write "***_", ces!*, "(", n,
        "):_Generation_of_new_equations_by_differentiation_possible."; terpri!* t;
      write "____Solvable_with_'on_allow_differentiation'"; terpri!* t; t >>

```

This code is used in section 55.

58. An algebraic expression is polynomial in a variable if the denominator does not depend on it and if the numerator is polynomial (we only have to check this if *polynomial\_check* is **on** ). The polynomial order we can obtain by simply reordering the numerator w.r.t. the variable involved.

```

lisp procedure get_polynomial_order(expression, variable);
  if ¬depends(denr(expression := simp expression), variable)
    ∧ (¬!*polynomial_check ∨ polynomialp(numr expression, variable)) then
    begin scalar kord!*;
      setkorder list !*a2k variable; expression := reorder numr expression;
      return if mvar expression = variable then ldeg expression else 0;
    end$

```

59. If none of the above methods can be applied, we cannot solve *equation*.

```

<Step 7: print a "Not solved" message 59> ≡
write ces!*, "(", n, "):_not_solved"; terpri!* t

```

This code is used in section 15.



**60. Additional tools.** The following procedures are meant for solving more equations at a time or solving “exceptional” equations, which need the least restrictive setting of the switches *coefficient\_check*, *polynomial\_check* or *allow\_differentiation*.

```
algebraic procedure integrate_equations(m, n);
  for i := m:n do integrate_equation(i)$
lisp operator integrate_exceptional_equation;
lisp procedure integrate_exceptional_equation(n);
  integrate_equation(n) where
    !*coefficient_check = nil,
    !*polynomial_check = nil,
    !*allow_differentiation = t$
```

**61.** For a system of equations which is not too difficult it may be possible to solve the system without intervention of the user. For such systems the procedure *auto\_solve* tries to solve a system automatically. If successful, it returns a message saying so, otherwise it returns the list of equation numbers left unsolved. The parameter *nr\_list* gives either the equation number or the list of equation numbers to be considered: the other equations may contain conditions which should only be considered when all the higher equations are solved, for instance when we solve using some kind of grading and solve the system degree by degree.

```
lisp operator auto_solve;
lisp procedure auto_solve nr_list;
  begin scalar total, old_total, to_do, unsolved, old_unsolved, stuck;
    total := old_total := get(ces!*, 'total_used');
    to_do := if fixp nr_list then list nr_list
      else if car nr_list = 'list then cdr nr_list
      else nr_list;
    while ¬stuck ∧ to_do do
      begin
        for each eq_nr in to_do do
          << integrate_equation eq_nr;
            if cadr assoc(list(ces!*, eq_nr), get(ces!*, 'kvalue')) ≠ 0 then unsolved := eq_nr . unsolved >>;
            total := get(ces!*, 'total_used');
            if total = old_total ∧ unsolved ∧ unsolved = old_unsolved then stuck := t
            else << old_unsolved := unsolved; to_do := reverse unsolved; unsolved := nil;
              to_do := append(for eq_nr := old_total + 1:total collect eq_nr, to_do); old_total := total >>
          end;
        if stuck then return 'list . reverse unsolved
      else << terpri(); write "Successful integration of all equations"; terpri() >>;
    end$
```

**62.** As a last set of tools, we shall give a procedure to print an equation together with all the functions occurring in it and their dependencies, and some procedures for showing and changing the properties of an equation set and a the functions/constants used.

As a side effect the procedure *show\_equation* will reassign the shown equation to its current value.

```

lisp operator show_equation;
lisp procedure show_equation n;
  begin scalar equation, total_used, function_list;
  if null(total_used := get(ces!*, 'total_used)) ∨ n > total_used then
    stop_with_error("SHOW_EQUATION: properly initialize", ces!*, nil, nil);
  if (equation := assoc(list(ces!*, n), get(ces!*, 'kvalue))) then
    begin equation := setk(list(ces!*, n), aeval cadr equation);
    varpri(equation, list('setk, mkquote list(ces!*, n), mkquote equation), 'only);
    function_list := get_recursive_kernels(numr simp equation, get(ces!*, 'function_list));
    if function_list then
      << terpri!* t; write "Functions occurring: "; terpri!* t;
      for each fn in function_list do
        << maprin(fn . get_dependencies_of(fn)); terpri!*(¬!*nat) >>>
      else terpri!* nil
    end
  end$
algebraic procedure show_equations(m, n);
  for i := m:n do show_equation i$

```

**63.**

```

lisp operator functions_used, put_functions_used, equations_used, put_equations_used;
lisp procedure functions_used function_name;
  list('list, get(function_name, 'even_used), get(function_name, 'odd_used))$
lisp procedure put_functions_used(function_name, even_used, odd_used);
  begin
  if ¬fixp even_used ∨ even_used < 0 ∨ ¬fixp odd_used ∨ odd_used < 0 then
    stop_with_error("PUT_FUNCTIONS_USED: used functions number invalid", nil, nil, nil);
  put(function_name, 'even_used, even_used); put(function_name, 'odd_used, odd_used);
  end$
lisp procedure equations_used;
  get(ces!*, 'total_used)$
lisp procedure put_equations_used(n);
  if ¬fixp n ∨ n < 0 then
    stop_with_error("PUT_EQUATIONS_USED: used equation number invalid", nil, nil, nil)
  else put(ces!*, 'total_used, n)$

```

64. There is one slight detail which we have not dealt with yet: in prolongation theory differentiation should act as a derivation on the arguments of a (eventually nested) commutator. In REDUCE 3.4 there is a hook which can take care of this situation. In the procedure *diffp*, which takes care of differentiation of standard powers, if this standard power is an operator kernel, the property *dfform* is checked for operator concerned. If this property has a value, it should be a function which takes care of the differentiation of such a standard power.

```
lisp operator df_acts_as_derivation_on;
lisp procedure df_acts_as_derivation_on operator_name;
  begin put (operator_name, 'dfform, 'df_as_derivation);
end$
```

65. The procedure *df\_as\_derivation* is quite straightforward: apply *df* to all the arguments of the operator, one at a time, leaving the other ones untouched.

```
lisp procedure df_as_derivation(kernel, variable, power);
  begin scalar left_part, right_part, argument, derivative;
  if power ≠ 1 then stop_with_error("DF_AS_DERIVATION:", kernel, "must occur linearly", nil);
  left_part := list operator_name_of kernel;
  right_part := arguments_of kernel;
  derivative := nil . 1;
  while right_part do
    << argument := first_element_of right_part;
    right_part := rest_of right_part;
    derivative := addsq(derivative, simp append(reverse left_part, list('df, argument,
      variable) . right_part));
    left_part := argument . left_part; >>;
  return derivative;
end$
```

66. In order to get nice output of some of the messages given by *integrate\_equation* we redefine the print function *listpri* for algebraic lists. Namely, we don't want algebraic lists to split over multiple lines in the messages we give. For this purpose, we introduce a fluid variable *listpri\_depth!\** which governs the depth for which algebraic lists are split along lines. The default value is the same as the value in the used in REDUCE.

```
(Lisp initializations 9) + ≡
  initialize_fluid(listpri_depth!*, 40)$
```

67. The following procedure can be used at algebraic level to change *listpri\_depth!\**.

```
lisp operator listlength$
lisp procedure listlength l;
  listpri_depth!* := l$
```

68. The definition of *listpri* is basically that of *inprint*, except that it decides when to split at the comma by looking at the size of the argument, using the global variable *listpri\_depth!\**.

```

symbolic procedure listpri l;
  begin scalar orig, split, u;
  u := l; l := cdr l; prin2!* get('!*lcbkt!*, 'prtch'); { Do it this way so table can change }
  orig := orig!*;
  orig!* := if posn!* < 18 then posn!* else orig!* + 3;
  if null l then go to b;
  split := treesizep(l, listpri_depth!*);
a: maprint(negnumberchk car l, 0); l := cdr l;
  if null l then go to b;
  oprin '!*comma!*';
  if split then terpri!* t;
  go to a;
b: prin2!* get('!*rcbkt!*, 'prtch'); orig!* := orig;
  return u
end$

```

69. The end of a REDUCE input file must be marked with **end**.

```

end;

```

**70. Index.** This section contains a cross reference index of all identifiers, together with the numbers of the modules in which they are used. Underlined entries correspond to module numbers where the identifier was declared.

*!\**: 4.  
*!allow\_differentiation*: 57, 60.  
*!a2k*: 58.  
*!coefficient\_check*: 22, 60.  
*!comma!\**: 68.  
*!ff2a*: 22, 38, 42.  
*!lcbkt!\**: 68.  
*!nat*: 19, 35, 57, 62.  
*!polynomial\_check*: 32, 48, 58, 60.  
*!rcbkt!\**: 68.  
*!...!*: 36.  
*absent\_variables*: 15, 29, 30, 31, 32, 39, 41.  
*action*: 19.  
*addsq*: 28, 50, 52, 65.  
*aeval*: 62.  
*allow\_differentiation*: 53, 60.  
*append*: 39, 40, 44, 61, 65.  
*argument*: 49, 65.  
*arguments\_of*: 2, 23, 47, 49, 50, 65.  
*assoc*: 16, 19, 22, 23, 52, 56, 61, 62.  
*assoc\_delete*: 19, 21, 42, 48.  
*assoc\_list*: 19.  
*atom*: 10, 11, 38.  
*auto\_solve*: 61.  
*banner*: 1, 5.  
*bracketname*: 9, 10, 11, 15, 23, 25, 27, 37, 38, 40.  
*broken\_list*: 36.  
*caddr*: 11.  
*caddr*: 2, 11.  
*cadr*: 2, 16, 56, 61, 62.  
*car*: 2, 17, 36, 52, 56, 61, 68.  
*cddr*: 56.  
*cdr*: 2, 17, 23, 36, 52, 61, 68.  
*ces!\**: 13, 14, 16, 19, 23, 24, 30, 35, 37, 38, 40, 48, 57, 59, 61, 62, 63.  
*change\_dimensions\_of*: 27.  
*change\_to\_algebraic\_mode*: 2, 5.  
*change\_to\_symbolic\_mode*: 2, 5.  
*check\_differentiation\_sequence*: 45, 47.  
*check\_polynomial\_integration*: 48, 49.  
*check\_valid\_function\_declaration*: 11, 12.  
*coefficient*: 50, 52.  
*coefficient\_check*: 18, 22, 39, 60.  
*coefficient\_name*: 23, 25, 28.  
*coefficient\_of*: 17, 22, 35, 52.  
*commutator\_function*: 56.  
*commutator\_functions*: 15, 39, 40, 44, 54, 55, 56.  
*constant\_operator*: 9, 10, 11, 23, 37.  
*constants\_list*: 15, 37, 40.  
*counter*: 57.  
*counter\_of*: 56, 57.  
*cur\_eq\_set!\**: 13.  
*decr*: 3.  
*define\_used*: 11.  
*delete*: 19, 28, 30, 39, 41, 44.  
*denominator*: 15, 16, 20, 21, 22, 38, 42, 43, 46, 48, 49.  
*denr*: 16, 49, 50, 58.  
*depend\_new\_coefficient*: 28.  
*dependency\_list*: 23, 28.  
*depends*: 34, 49, 58.  
*depl!\**: 21, 23, 28, 42, 48, 56.  
*depl\_entry*: 23.  
*derivative*: 65.  
*df*: 17, 18, 20, 24, 37, 38, 39, 40, 43, 45, 53, 65.  
*df\_acts\_as\_derivation\_on*: 64.  
*df\_as\_derivation*: 64, 65.  
*df\_function*: 23, 24, 27, 44.  
*df\_functions*: 15, 39, 40, 44.  
*df\_kernel*: 21, 46, 48.  
*df\_list*: 15, 20, 21, 37, 38, 40, 43, 46, 48.  
*df\_sequence*: 50, 51.  
*df\_term*: 23, 24, 40, 47, 49, 50, 56, 57.  
*df\_terms*: 15, 40, 43, 46, 47, 54, 55, 56.  
*dfform*: 64.  
*differentiations\_list*: 55, 57.  
*diffp*: 64.  
*domainp*: 22, 34.  
*empty\_function*: 56.  
*entry*: 56, 57.  
*eq\_nr*: 61.  
*equ*: 13.  
*equation*: 15, 16, 17, 20, 29, 30, 31, 32, 35, 37, 38, 39, 42, 43, 46, 48, 53, 54, 55, 57, 59, 62.  
*equations\_list*: 31, 35.  
*equations\_used*: 63.  
*even\_dimension*: 27.  
*even\_used*: 10, 11, 12, 23, 25, 27, 28, 63.  
*expr\_1*: 2.  
*expr\_2*: 2.  
*expression*: 34, 58.  
*ext\_mksq*: 28.  
*find\_solvable\_kernel*: 18, 21, 22, 39, 42, 48.  
*first\_argument\_of*: 2, 11, 21, 24, 40, 47, 48, 56.

*first\_element\_of*: 2, 10, 11, 21, 22, 26, 45, 48, 49, 51, 65.  
*first\_solvable\_kernel*: 22.  
*fixp*: 10, 11, 24, 26, 45, 49, 51, 61, 63.  
*flag*: 4.  
*flag\_function*: 56.  
*fluid*: 4.  
*fluid\_name*: 4.  
*fn*: 62.  
*forbidden\_functions*: 46, 47.  
*function*: 2, 30.  
*function\_list*: 10, 12, 15, 24, 30, 37, 40, 62.  
*function\_name*: 10, 12, 63.  
*function\_number*: 23, 24, 27, 28.  
*function\_operator*: 9.  
*function\_specification*: 12.  
*functions\_and\_constants\_list*: 15, 40, 42.  
*functions\_used*: 63.  
*get*: 11, 16, 23, 24, 25, 27, 30, 35, 37, 40, 57, 61, 62, 63, 68.  
*get\_dependencies\_of*: 23, 30, 39, 44, 47, 56, 62.  
*get\_polynomial\_order*: 57, 58.  
*get\_recursive\_kernels*: 30, 40, 62.  
*global*: 4.  
*global\_name*: 4.  
*homogeneous\_integration\_of*: 21, 23, 50.  
*identifier*: 2.  
*identity\_function*: 56.  
*idp*: 10, 11, 14.  
*incr*: 3, 28, 35, 57.  
*independent\_part\_of*: 17, 21, 35, 37, 40, 50.  
*inhomogeneous\_integration\_of*: 48, 50.  
*inhomogeneous\_term*: 46, 48, 50.  
*initialize\_equations*: 9.  
*initialize\_equations1*: 9, 10.  
*initialize\_fluid*: 4, 66.  
*initialize\_global*: 4, 13.  
*inprint*: 68.  
*int\_factor*: 50, 52.  
*int\_sequence*: 50, 51, 52.  
*integrate\_equation*: 15, 60, 61, 66.  
*integrate\_equations*: 60.  
*integrate\_exceptional\_equation*: 60.  
*integration*: 52.  
*integration\_list*: 23, 26.  
*integration\_term*: 49.  
*integration\_terms*: 50, 52.  
*integration\_variable*: 23, 26, 28.  
*integration\_variables*: 15, 43, 44, 45, 46, 47, 49, 50, 51.

*kc\_list*: 22.  
*kc\_list\_of*: 17, 21, 29, 37, 40.  
*kc\_pair*: 22.  
*kernel*: 19, 23, 28, 34, 39, 44, 56, 65.  
*kernel\_list*: 22, 56.  
*kernel\_of*: 17, 21, 22, 29, 40.  
*kernel\_selector*: 56.  
*kord!\**: 58.  
*kvalue*: 16, 61, 62.  
*lc*: 34.  
*ldeg*: 58.  
*left\_part*: 65.  
*length*: 10, 11, 21, 23, 35, 38, 39, 41, 44, 47, 48.  
*liebracket*: 23.  
*linear\_function*: 40, 56, 57.  
*linear\_functions*: 15, 39, 40, 44, 54, 55, 56.  
*linear\_functions\_list*: 15, 37, 38, 40.  
*linear\_solve*: 48, 57.  
*linear\_solve\_and\_assign*: 42.  
*list*: 2, 10, 11, 16, 21, 28, 32, 35, 36, 37, 40, 47, 49, 57, 58, 61, 62, 63, 65.  
*listlength*: 67.  
*listpri*: 66, 68.  
*listpri\_depth!\**: 15, 66, 67, 68.  
*main\_variable*: 34.  
*maprin*: 19, 35, 38, 57, 62.  
*maprint*: 68.  
*member*: 24, 40, 45, 47.  
*message*: 2.  
*mk!\*sq*: 28, 35, 48, 50, 57.  
*mkquote*: 62.  
*mksq*: 28, 52.  
*msgpri*: 2.  
*multi\_split\_form*: 35, 50.  
*multsq*: 28, 50, 52.  
*mvar*: 34, 58.  
*negnumberchk*: 68.  
*new\_coefficient*: 28.  
*new\_dependency\_list*: 23, 28.  
*new\_switch*: 4, 18, 33, 53.  
*not\_a\_number\_message\_for*: 19, 21, 42, 48.  
*nr\_list*: 61.  
*nr\_of\_integrations*: 50, 51.  
*nr\_of\_items*: 36.  
*nr\_of\_variables*: 15, 39, 41, 43, 44, 46, 47.  
*null*: 16, 21, 45, 62, 68.  
*nullify\_equation*: 16, 19, 35, 38.  
*number\_of\_integrations*: 23, 26, 27, 28.  
*number\_of\_occurences*: 56.  
*numberp*: 22.

- numerator*: 49.
- numr*: 16, 49, 50, 58, 62.
- odd\_dimension*: 27.
- odd\_used*: 10, 11, 12, 23, 25, 27, 28, 63.
- ok*: 49.
- old\_total*: 61.
- old\_unsolved*: 61.
- only*: 62.
- op\_list*: 11.
- op\_name*: 11.
- operator\_name*: 9, 10, 11, 12, 14, 64.
- operator\_name\_of*: 2, 10, 11, 23, 24, 40, 65.
- oprin*: 68.
- orig*: 68.
- orig!\**: 68.
- origin*: 56, 57.
- origin\_of*: 56, 57.
- partial\_list*: 35, 36, 42, 57.
- pc\_list*: 50.
- pc\_list\_of*: 29, 35, 50.
- pc\_pair*: 35.
- polynomial\_check*: 32, 33, 43, 58, 60.
- polynomial\_order*: 55, 57.
- polynomial\_variables*: 31, 32, 35.
- polynomialp*: 32, 34, 49, 58.
- posn!\**: 68.
- power*: 28, 52, 65.
- powers*: 50, 52.
- powers\_of*: 29, 52.
- present\_functions\_list*: 15, 30.
- present\_variables*: 15, 39, 41, 43, 44, 54, 55, 56, 57.
- printed\_list*: 36.
- prin2!\**: 68.
- prtch*: 68.
- psopfn*: 9.
- put*: 9, 10, 11, 12, 35, 57, 63, 64.
- put\_equations\_used*: 63.
- put\_functions\_used*: 63.
- put\_used\_dimensions*: 11, 12, 28.
- recursive\_functions\_list*: 39.
- red*: 34.
- rederr*: 10, 14.
- reinitialize\_present\_variables*: 56.
- relation\_analysis*: 38.
- reorder*: 58.
- rest\_of*: 2, 10, 11, 12, 22, 23, 26, 45, 47, 49, 50, 51, 65.
- reval*: 10, 11.
- reverse*: 61, 65.
- right\_part*: 65.
- rplacd*: 56.
- second\_argument\_of*: 2.
- sequence*: 45.
- setk*: 16, 21, 35, 48, 57, 62.
- setkorder*: 58.
- show\_equation*: 62.
- show\_equations*: 62.
- simp*: 49, 50, 58, 62, 65.
- simp!\**: 16.
- simpdf*: 57.
- skip\_list*: 2, 10.
- solution*: 23, 28, 50, 52.
- solution\_term*: 50, 52.
- solvable\_kernel*: 15, 21, 38, 42, 46, 48.
- solvable\_kernels*: 15, 21, 39, 42, 46, 47, 48.
- solved*: 15, 16, 20, 29, 38, 42, 43, 54.
- specification*: 10, 11.
- specification\_list*: 10, 11, 12.
- split*: 68.
- split\_equation\_polynomially*: 29, 31, 35.
- split\_form*: 17, 20, 37.
- stop\_with\_error*: 2, 11, 16, 24, 62, 63, 65.
- string\_1*: 2.
- string\_2*: 2.
- stuck*: 61.
- subs2*: 28, 50.
- successful\_message\_for*: 19, 21, 42, 48.
- switch*: 4.
- switch\_name*: 4.
- term*: 52.
- terpri*: 5, 61.
- terpri!\**: 15, 16, 19, 35, 38, 48, 57, 59, 62, 68.
- to\_do*: 61.
- total*: 61.
- total\_used*: 9, 10, 15, 16, 29, 31, 35, 54, 55, 57, 61, 62, 63.
- treesizep*: 68.
- try\_a\_differentiation*: 54, 55.
- try\_a\_homogeneous\_integration*: 20, 21.
- try\_an\_inhomogeneous\_integration*: 43, 46.
- unsolved*: 61.
- update\_variable*: 56.
- update\_variables\_using*: 56.
- use\_equations*: 14.
- value*: 4.
- variable*: 30, 32, 41, 44, 49, 50, 51, 52, 56, 58, 65.
- variable\_list*: 9, 10, 15, 30, 39, 41, 45.
- variable\_of*: 56, 57.
- varpri*: 62.
- write*: 5, 16, 19, 35, 38, 48, 57, 59, 61, 62.

(Check and initialize *constant\_operator* 11) Used in section 10.  
 (Check and initialize *function\_list* 12) Used in section 10.  
 (Check and possibly enlarge dimensions of *bracketname* 27) Used in section 23.  
 (Check if *df\_term* can be integrated, find *df\_function* and *function\_number* 24) Used in section 23.  
 (Construct *df\_terms*, *df\_functions*, *linear\_functions* and *commutator\_functions* 40) Used in section 39.  
 (Count the number of occurrences of all *present\_variables* 56) Used in section 55.  
 (Find a *solvable\_kernel*, check the *inhomogeneous\_term* and possibly integrate 48) Used in section 46.  
 (Find the equation to be integrated 16) Used in section 15.  
 (Find the integrable *df\_terms* 47) Used in section 46.  
 (Find the next *integration\_variable* and *number\_of\_integrations* 26) Used in sections 23 and 28.  
 (Find the possible *integration\_variables* 44) Used in section 43.  
 (Find the *integration\_variables* and *int\_sequence* 51) Used in section 50.  
 (Find the *polynomial\_variables* and test for polynomial behaviour 32) Used in section 31.  
 (Find *present\_functions\_list* and the *absent\_variables* 30) Used in section 29.  
 (Get *even\_used*, *odd\_used* and if necessary *bracketname* 25) Used in section 23.  
 (Get *present\_variables* and *nr\_of\_variables* 41) Used in section 39.  
 (If possible and allowed, generate new equations 57) Used in section 55.  
 (If possible, split up *equation* into smaller equations 35) Used in section 31.  
 (Lisp initializations 9, 13, 18, 33, 53, 66) Used in section 5.  
 (Perform the inhomogeneous integration of the numerator of *inhomogeneous\_term* 52) Used in section 50.  
 (Perform the integration 28) Used in section 23.  
 (Solve *equation* if it is a Lie expression 38) Used in section 37.  
 (Step 1: search for homogeneous integration 20) Used in section 15.  
 (Step 2: search for polynomial behaviour 29) Used in section 15.  
 (Step 3: search for a Lie relation 37) Used in section 15.  
 (Step 4: search for a solvable function 39) Used in section 15.  
 (Step 5: search for inhomogeneous integration 43) Used in section 15.  
 (Step 6: search for a useful differentiation 54) Used in section 15.  
 (Step 7: print a “Not solved” message 59) Used in section 15.  
 (Try to solve a function 42) Used in section 39.